We study a satellite in a circular equatorial orbit. See T. Kailath, *Linear Systems*, Prentice-Hall, 1980, p. 60 and p. 145, and H. Mounier, *Propriétés structurelles des systèmes linéaires à retards: aspects théoriques et pratiques*, PhD Thesis, University of Orsay, France, 1995, p. 6, p. 11 and p. 17.

```
> with(Ore_algebra):
> with(OreModules):
```

We define the Weyl algebra $Alg = A_1$, where Dt acts as differentiation w.r.t. time t. Note that we have to declare the parameters ω (angular velocity), m (mass of the satellite), r (radius component in the polar coordinates), a and b (parameters specifying the thrust) of the system in the definition of the Ore algebra:

```
> Alg := DefineOreAlgebra(diff=[Dt,t], polynom=[t], comm=[omega,m,r,a,b]):
```

The linearized ordinary differential equations for the satellite in a circular orbit are given by the following matrix R. These equations describe the motion of the satellite in the equatorial plane, where the fifth and the sixth column of R incorporate the controls $u1$, $u2$ which represent radial thrust resp. tangential thrust caused by rocket engines (see Kailath, 1980, p. 60 and p. 145).

```
> Rab := evalm([[Dt,-1,0,0,0,0], [-3*omega^2,Dt,0,-2*omega*r,-a/m,0],
> [0,0,Dt,-1,0,0], [0,2*omega/r,0,Dt,0,-b/(m*r)]]);
```

$$Rab := \begin{bmatrix} Dt & -1 & 0 & 0 & 0 & 0 \\ -3\,\omega^2 & Dt & 0 & -2\,\omega\,r & -\dfrac{a}{m} & 0 \\ 0 & 0 & Dt & -1 & 0 & 0 \\ 0 & \dfrac{2\,\omega}{r} & 0 & Dt & 0 & -\dfrac{b}{m\,r} \end{bmatrix}$$

We define the formal adjoint R_adj of R:

```
> Rab_adj := Involution(Rab, Alg);
```

$$Rab\_adj := \begin{bmatrix} -Dt & -3\,\omega^2 & 0 & 0 \\ -1 & -Dt & 0 & \dfrac{2\,\omega}{r} \\ 0 & 0 & -Dt & 0 \\ 0 & -2\,\omega\,r & -1 & -Dt \\ 0 & -\dfrac{a}{m} & 0 & 0 \\ 0 & 0 & 0 & -\dfrac{b}{m\,r} \end{bmatrix}$$

Let us compute the first extension module ext^1 with values in Alg of the Alg-module N associated with R_adj:

```
> Extab := Exti(Rab_adj, Alg, 1);
```

$$Extab := \left[\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} -3\,m\,\omega^2 & Dt\,m & 0 & -2\,\omega\,r\,m & -a & 0 \\ Dt & -1 & 0 & 0 & 0 & 0 \\ 0 & 2\,m\,\omega & 0 & m\,r\,Dt & 0 & -b \\ 0 & 0 & Dt & -1 & 0 & 0 \end{bmatrix},\right.$$
$$\left.\begin{bmatrix} b\,a & 0 \\ b\,a\,Dt & 0 \\ 0 & b\,a \\ 0 & b\,a\,Dt \\ -3\,b\,m\,\omega^2 + Dt^2\,b\,m & -2\,Dt\,b\,\omega\,r\,m \\ 2\,a\,Dt\,m\,\omega & a\,Dt^2\,m\,r \end{bmatrix}\right]$$

Since *Ext1*[1] is the identity matrix, we conclude that ext^1 of N is the zero module. Hence, the module M which is associated with the system R is torsion-free. It follows that the system is controllable and, equivalently, parametrizable. A parametrization of R is given in *Ext1*[3]. Of course, a necessary condition for *Ext1*[3] being a parametrization is that (R o *Ext1*[3]) $= 0$:

```
> Mult(Rab, Extab[3], Alg);
```

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \\ 0 & 0 \\ 0 & 0 \end{bmatrix}$$

Equivalently, a parametrization of the system can be computed by using the following command:

```
> Parametrization(Rab, Alg);
```

$$\begin{bmatrix} b\,a\,\xi_1(t) \\ b\,a\,(\frac{d}{dt}\,\xi_1(t)) \\ b\,a\,\xi_2(t) \\ b\,a\,(\frac{d}{dt}\,\xi_2(t)) \\ -b\,m\,(3\,\omega^2\,\xi_1(t) - (\frac{d^2}{dt^2}\,\xi_1(t)) + 2\,\omega\,r\,(\frac{d}{dt}\,\xi_2(t))) \\ a\,m\,(2\,\omega\,(\frac{d}{dt}\,\xi_1(t)) + r\,(\frac{d^2}{dt^2}\,\xi_2(t))) \end{bmatrix}$$

The coefficients in the equations of the system lie in the polynomial ring with one variable Dt and with coefficients that are rational functions in ω, m, r, a, b with real coefficients. Since this polynomial ring is a principal ideal domain (namely, every ideal is generated by a single element), we know that torsion-freeness of the module M which is associated with the system R actually implies *freeness*, i.e., system R is *flat*. Hence, we can compute a left-inverse of the parametrization and get a *flat output* of the system:

```
> Sab := LeftInverse(Extab[3], Alg);
```

$$Sab := \begin{bmatrix} \frac{1}{b\,a} & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{1}{b\,a} & 0 & 0 & 0 \end{bmatrix}$$

Therefore, $(\xi 1 : \xi 2)^T = Sab\ (x1 : x2 : x3 : x4 : u1 : u2)^T$ is a flat output of the system which satisfies $(x1 : x2 : x3 : x4 : u1 : u2)^T = Extab[3]\ (\xi 1 : \xi 2)^T$. Let us notice that this flat output exists only if $ab \neq 0$.

Let us remember that the full row-rank matrix R admits a right-inverse if and only if the module which is associated with it is projective. By the theorem of Quillen-Suslin, for modules over commutative polynomial rings, projectiveness is the same as freeness. So, M is projective which we could have discovered by computing a right-inverse of R:

```
> RightInverse(Rab, Alg);
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ -\frac{Dt\,m}{a} & -\frac{m}{a} & \frac{2\,\omega\,r\,m}{a} & 0 \\ -\frac{2\,\omega\,m}{b} & 0 & -\frac{Dt\,m\,r}{b} & -\frac{m\,r}{b} \end{bmatrix}$$

Let us compute a Brunovský canonical form for the system defined by R in the case where $ab \neq 0$.

```
> B := Brunovsky(Rab, Alg);
```

$$B := \begin{bmatrix} \frac{1}{b\,a} & 0 & 0 & 0 & 0 & 0 \\ 0 & \frac{1}{b\,a} & 0 & 0 & 0 & 0 \\ \frac{3\,\omega^2}{b\,a} & 0 & 0 & \frac{2\,\omega\,r}{b\,a} & \frac{1}{b\,m} & 0 \\ 0 & 0 & \frac{1}{b\,a} & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{1}{b\,a} & 0 & 0 \\ 0 & -\frac{2\,\omega}{b\,a\,r} & 0 & 0 & 0 & \frac{1}{m\,a\,r} \end{bmatrix}$$

Therefore, using the following change of variables

```
> evalm([[z[1](t)],[z[2](t)],[v[1](t)],[z[3](t)],[z[4](t)],[v[2](t)]])=
> ApplyMatrix(B, [seq(x[i](t),i=1..4),u1(t),u2(t)], Alg);
```

$$\begin{bmatrix} z_1(t) \\ z_2(t) \\ v_1(t) \\ z_3(t) \\ z_4(t) \\ v_2(t) \end{bmatrix} = \begin{bmatrix} \frac{x_1(t)}{b\,a} \\ \frac{x_2(t)}{b\,a} \\ \frac{3\,\omega^2\,x_1(t)}{b\,a} + \frac{2\,\omega\,r\,x_4(t)}{b\,a} + \frac{\mathrm{u1}(t)}{b\,m} \\ \frac{x_3(t)}{b\,a} \\ \frac{x_4(t)}{b\,a} \\ -\frac{2\,\omega\,x_2(t)}{b\,a\,r} + \frac{\mathrm{u2}(t)}{m\,a\,r} \end{bmatrix}$$

we obtain the following Brunovský canonical form:

```
> E := Elimination(linalg[stackmatrix](B, Rab),
> [seq(x[i],i=1..4),u1,u2], [z[1],z[2],v[1],z[3],z[4],v[2],0,0,0,0], Alg):
> ApplyMatrix(E[1], [seq(x[i](t),i=1..4),u1(t),u2(t)], Alg)=
> ApplyMatrix(E[2], [[z[1](t)],[z[2](t)],[v[1](t)],[z[3](t)],[z[4](t)],[v[2](t)]],
> Alg);
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ \mathrm{u2}(t) \\ \mathrm{u1}(t) \\ x_4(t) \\ x_3(t) \\ x_2(t) \\ x_1(t) \end{bmatrix} = \begin{bmatrix} -(\frac{d}{dt}\,z_4(t)) + v_2(t) \\ -(\frac{d}{dt}\,z_3(t)) + z_4(t) \\ -(\frac{d}{dt}\,z_2(t)) + v_1(t) \\ -(\frac{d}{dt}\,z_1(t)) + z_2(t) \\ 2\,a\,\omega\,m\,z_2(t) + a\,m\,r\,v_2(t) \\ -3\,b\,\omega^2\,m\,z_1(t) + b\,m\,v_1(t) - 2\,b\,\omega\,r\,m\,z_4(t) \\ b\,a\,z_4(t) \\ b\,a\,z_3(t) \\ b\,a\,z_2(t) \\ b\,a\,z_1(t) \end{bmatrix}$$

Let us consider the case where $a = 0$ and $b = 1$, i.e., the case where we only have a tangential thrust. Then, the system is defined by the following matrix:

```
> R01 := linalg[submatrix](subs(a=0,b=1,evalm(Rab)), 1..4, [1,2,3,4,6]);
```

$$R01 := \begin{bmatrix} Dt & -1 & 0 & 0 & 0 \\ -3\,\omega^2 & Dt & 0 & -2\,\omega\, r & 0 \\ 0 & 0 & Dt & -1 & 0 \\ 0 & \frac{2\,\omega}{r} & 0 & Dt & -\frac{1}{m\,r} \end{bmatrix}$$

The formal adjoint *R01_ adj* of *R01* is defined by:

```
> R01_adj := Involution(R01, Alg);
```

$$R01\_adj := \begin{bmatrix} -Dt & -3\,\omega^2 & 0 & 0 \\ -1 & -Dt & 0 & \frac{2\,\omega}{r} \\ 0 & 0 & -Dt & 0 \\ 0 & -2\,\omega\, r & -1 & -Dt \\ 0 & 0 & 0 & -\frac{1}{m\,r} \end{bmatrix}$$

Let us check whether or not the new system is controllable.

```
> Ext01 := Exti(R01_adj, Alg, 1);
```

$$Ext01 := \left[\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} -3\,\omega^2 & Dt & 0 & -2\,\omega\, r & 0 \\ Dt & -1 & 0 & 0 & 0 \\ 0 & 2\,\omega\, m & 0 & m\, r\, Dt & -1 \\ 0 & 0 & Dt & -1 & 0 \end{bmatrix}, \begin{bmatrix} 2\,Dt\,\omega\, r \\ 2\,\omega\, Dt^2\, r \\ -3\,\omega^2 + Dt^2 \\ Dt^3 - 3\,Dt\,\omega^2 \\ Dt^4\, m\, r + Dt^2\,\omega^2\, m\, r \end{bmatrix} \right]$$

We obtain that the *Alg*-module associated with *R01* is torsion-free, and thus, the system is controllable. Moreover, a parametrization of the system is given by *Ext01*[3] or, equivalently, by:

```
> Parametrization(R01, Alg);
```

$$\begin{bmatrix} 2\,\omega\, r\,(\frac{d}{dt}\,\xi_1(t)) \\ 2\,\omega\, r\,(\frac{d^2}{dt^2}\,\xi_1(t)) \\ -3\,\omega^2\,\xi_1(t) + (\frac{d^2}{dt^2}\,\xi_1(t)) \\ (\frac{d^3}{dt^3}\,\xi_1(t)) - 3\,\omega^2\,(\frac{d}{dt}\,\xi_1(t)) \\ m\, r\,(\omega^2\,(\frac{d^2}{dt^2}\,\xi_1(t)) + (\frac{d^4}{dt^4}\,\xi_1(t))) \end{bmatrix}$$

Using the fact that the system is time-invariant, we obtain that the *Alg*-module associated with *R01* is free, and thus, the system is flat. A flat output is obtain by computing a left-inverse of the parametrization *Ext01*[3].

```
> S01 := LeftInverse(Ext01[3], Alg);
```

$$S01 := \begin{bmatrix} 0 & \frac{1}{6\, r\,\omega^3} & -\frac{1}{3\,\omega^2} & 0 & 0 \end{bmatrix}$$

```
> simplify(Mult(S01, Ext01[3], Alg));
```

$$\begin{bmatrix} 1 \end{bmatrix}$$

Therefore, $\xi = S01\;(x1 : x2 : x3 : x4 : u2)^T$ is a flat output of the system which satisfies

$$(x1 : x2 : x3 : x4 : u2)^T = Ext01[3]\,\xi.$$

Now, let us turn to the case where $a = 1$ and $b = 0$, i.e., to the case where there is only a radial thrust. Then, the system is defined by the following matrix:

```
> R10 := linalg[submatrix](subs(a=1,b=0,evalm(Rab)), 1..4, 1..5);
```

$$R10 := \begin{bmatrix} Dt & -1 & 0 & 0 & 0 \\ -3\,\omega^2 & Dt & 0 & -2\,\omega\,r & -\dfrac{1}{m} \\ 0 & 0 & Dt & -1 & 0 \\ 0 & \dfrac{2\,\omega}{r} & 0 & Dt & 0 \end{bmatrix}$$

Its formal adjoint *R10_ adj* is defined by:

```
> R10_adj := Involution(R10, Alg);
```

$$R10\_adj := \begin{bmatrix} -Dt & -3\,\omega^2 & 0 & 0 \\ -1 & -Dt & 0 & \dfrac{2\,\omega}{r} \\ 0 & 0 & -Dt & 0 \\ 0 & -2\,\omega\,r & -1 & -Dt \\ 0 & -\dfrac{1}{m} & 0 & 0 \end{bmatrix}$$

Let us check whether or not the system defined by *R10* is controllable.

```
> Ext01 := Exti(R10_adj, Alg, 1);
```

$$Ext01 := \left[\begin{bmatrix} Dt & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & Dt & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 2\,\omega & 0 & 0 & r & 0 \\ 0 & 2\,\omega & 0 & Dt\,r & 0 \\ 0 & 2\,m\,Dt & 0 & -\omega\,r\,m & -2 \\ 0 & 0 & Dt & -1 & 0 \end{bmatrix}, \begin{bmatrix} -Dt\,r \\ -Dt^2\,r \\ 2\,\omega \\ 2\,Dt\,\omega \\ -Dt\,\omega^2\,r\,m - Dt^3\,r\,m \end{bmatrix} \right]$$

The first matrix of *Ext01* is not the identity matrix, and thus, there exist some torsion elements in the *Alg*-module associated with *R01*. Let us compute them.

```
> TorsionElements(R10, [x1(t),x2(t),x3(t),x4(t),u1(t)], Alg);
```

$$\left[\begin{bmatrix} \frac{d}{dt}\,\theta_1(t) = 0 \\ \frac{d}{dt}\,\theta_3(t) = 0 \end{bmatrix}, \begin{bmatrix} \theta_1(t) = 2\,\omega\,\mathrm{x1}(t) + r\,\mathrm{x4}(t) \\ \theta_3(t) = 2\,m\,(\frac{d}{dt}\,\mathrm{x2}(t)) - \omega\,r\,m\,\mathrm{x4}(t) - 2\,\mathrm{u1}(t) \end{bmatrix} \right]$$

Then, the autonomous elements of the system are defined by:

```
> AutonomousElements(R10, [x1(t),x2(t),x3(t),x4(t),u1(t)], Alg);
```

$$\left[\begin{bmatrix} 3\,\omega\,m\,\theta_1(t) - \theta_2(t) = 0 \\ \frac{d}{dt}\,\theta_2(t) = 0 \end{bmatrix}, \begin{bmatrix} \theta_1 = \dfrac{\_C1}{3\,\omega\,m} \\ \theta_2 = \_C1 \end{bmatrix}, \begin{bmatrix} \theta_1 = 2\,\omega\,\mathrm{x1}(t) + r\,\mathrm{x4}(t) \\ \theta_2 = 2\,m\,(\frac{d}{dt}\,\mathrm{x2}(t)) - \omega\,r\,m\,\mathrm{x4}(t) - 2\,\mathrm{u1}(t) \end{bmatrix} \right]$$

In particular, the system is not controllable. A first integral of motion of the system is defined by:

```
> FirstIntegral(R10, [x1(t),x2(t),x3(t),x4(t),u1(t)], Alg);
```

$$\frac{1}{2}\,\frac{\_C1\,(2\,\omega\,\mathrm{x1}(t) + r\,\mathrm{x4}(t))}{\omega}$$

We let the reader check by himself that the time-derivative of the above first integral of motion is 0 modulo the system equations.

Finally, let us point out that the controllable part of the system is defined by the matrix *Ext01*[2] and it is parametrized by *Ext01*[3].

Following (Mounier, 1995), we modify the description of the control of the satellite in the system. If the rocket engines are commanded from the earth, then, due to transmission time, a constant time-delay occurs in the system.

Hence, we enlarge the above Ore algebra by a shift operator δ:

```
> Alg2 := DefineOreAlgebra(diff=[Dt,t], dual_shift=[delta,s],
> polynom=[t,s], comm=[omega,m,r,a,b], shift_action=[delta,t]):
```

The system matrix is given as follows:

```
> R2 := evalm([[Dt,-1,0,0,0,0], [-3*omega^2,Dt,0,-2*omega*r,-a*delta/m,0],
> [0,0,Dt,-1,0,0], [0,2*omega/r,0,Dt,0,-b*delta/(m*r)]]);
```

$$R2 := \begin{bmatrix} Dt & -1 & 0 & 0 & 0 & 0 \\ -3\,\omega^2 & Dt & 0 & -2\,\omega\,r & -\dfrac{a\,\delta}{m} & 0 \\ 0 & 0 & Dt & -1 & 0 & 0 \\ 0 & \dfrac{2\,\omega}{r} & 0 & Dt & 0 & -\dfrac{b\,\delta}{m\,r} \end{bmatrix}$$

We define a formal adjoint *R2_adj* of *R2* using an involution of *Alg2*:

```
> R2_adj := Involution(R2, Alg2);
```

$$R2\_adj := \begin{bmatrix} -Dt & -3\,\omega^2 & 0 & 0 \\ -1 & -Dt & 0 & \dfrac{2\,\omega}{r} \\ 0 & 0 & -Dt & 0 \\ 0 & -2\,\omega\,r & -1 & -Dt \\ 0 & \dfrac{a\,\delta}{m} & 0 & 0 \\ 0 & 0 & 0 & \dfrac{b\,\delta}{m\,r} \end{bmatrix}$$

We check controllability and parametrizability of the system by applying *Exti* to *R2_adj*:

```
> Ext1 := Exti(R2_adj, Alg2, 1);
```

$$Ext1 := \left[\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} -3\,m\,\omega^2 & Dt\,m & 0 & -2\,\omega\,r\,m & -a\,\delta & 0 \\ Dt & -1 & 0 & 0 & 0 & 0 \\ 0 & 2\,m\,\omega & 0 & m\,r\,Dt & 0 & -b\,\delta \\ 0 & 0 & Dt & -1 & 0 & 0 \end{bmatrix}, \right.$$
$$\left.\begin{bmatrix} b\,a\,\delta & 0 \\ b\,a\,\delta\,Dt & 0 \\ 0 & b\,a\,\delta \\ 0 & b\,a\,\delta\,Dt \\ -3\,b\,m\,\omega^2 + Dt^2\,b\,m & -2\,Dt\,b\,\omega\,r\,m \\ 2\,a\,Dt\,m\,\omega & a\,Dt^2\,m\,r \end{bmatrix}\right]$$

Since *Ext1*[1] is the identity matrix, we find that the first extension module with values in *Alg2* of the *Alg2*-module N which is associated with *R2_adj* is generically the zero module. Equivalently, the system is generically controllable and parametrizable. A parametrization of the system is given in *Ext1*[3]. Equivalently, we can directly obtain the parametrization of the system by using the following command:

```
> Parametrization(R2, Alg2);
```

$$\begin{bmatrix} b\,a\,\xi_1(t-1) \\ b\,a\,\mathrm{D}(\xi_1)(t-1) \\ b\,a\,\xi_2(t-1) \\ b\,a\,\mathrm{D}(\xi_2)(t-1) \\ b\,m\,(\mathrm{D}^{(2)})(\xi_1)(t) - 3\,b\,\omega^2\,m\,\xi_1(t) - 2\,b\,\omega\,r\,m\,\mathrm{D}(\xi_2)(t) \\ 2\,a\,\omega\,m\,\mathrm{D}(\xi_1)(t) + a\,m\,r\,(\mathrm{D}^{(2)})(\xi_2)(t) \end{bmatrix}$$

This parametrization is actually a minimal one (namely, it involves the minimal number of free functions) because we obtain the same parametrization using *MinimalParametrization*:

```
> MinimalParametrization(R2, Alg2);
```

$$\begin{bmatrix} b\,a\,\delta & 0 \\ b\,a\,\delta\,Dt & 0 \\ 0 & b\,a\,\delta \\ 0 & b\,a\,\delta\,Dt \\ -3\,b\,m\,\omega^2 + Dt^2\,b\,m & -2\,Dt\,b\,\omega\,r\,m \\ 2\,a\,Dt\,m\,\omega & a\,Dt^2\,m\,r \end{bmatrix}$$

We continue to study the structural properties of the system by examining the algebraic properties of the *Alg2*-module M which is associated with *R2*. The next step is to compute the second extension module with values in *Alg2* of N:

```
> Ext2 := Exti(R2_adj, Alg2, 2);
```

$$\mathit{Ext2} := \left[\begin{bmatrix} \delta & 0 \\ Dt\,\omega^2 + Dt^3 & 0 \\ 0 & \delta \\ 0 & Dt^2\,\omega^2 + Dt^4 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \mathrm{SURJ}(2)\right]$$

Since *Ext2*[1] is not an identity matrix, we see that ext^2 of N is different from zero. Hence, M is not projective which also implies that M is not free. So, the satellite is not a flat system. As already mentioned above, M is a projective *Alg2*-module if and only if the full row rank matrix *R2* admits a right-inverse. We conclude that a right-inverse of *R2* does not exist:

```
> RightInverse(R2, Alg2);
```

$$[]$$

Since the torsion-free degree $\mathrm{i}(M)$ of M is equal to 1, we can find a polynomial π in the variable δ such that the system is π-free:

```
> PiPolynomial(R2, Alg2, [delta]);
```

$$[\delta]$$

By definition of the π-polynomial (Mounier, 1995), this means that if we introduce the time-advance operator in the system of the satellite, then it becomes a flat system. Hence, the module M associated with this system is a free module (over the Ore algebra which is obtained by adjoining the advance operator δ^{-1} to *Alg2*), and we are going to find a basis for this module using *LocalLeftInverse*:

```
> S := LocalLeftInverse(Ext1[3], [delta], Alg2);
```

$$S := \begin{bmatrix} 0 & 0 & -\frac{r\,Dt\,(Dt^2+4\,\omega^2)}{6\,\delta\,a\,\omega^3\,b} & 0 & -\frac{1}{3\,\omega^2\,b\,m} & \frac{Dt}{6\,a\,\omega^3\,m} \\ 0 & 0 & \frac{1}{\delta\,b\,a} & 0 & 0 & 0 \end{bmatrix}$$

We obtain a left-inverse S of the parametrization *Ext1*[3] of the system, where we admit δ in the denominators, i.e., we allow the time-advance operator.

```
> Mult(S, Ext1[3], Alg2);
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Hence, $(z_1 : z_2)^T = S\ (x_1 : x_2 : x_3 : x_4 : u_1 : u_2)^T$ is a basis of the $Alg2[\delta^{-1}]$-module M_2, and thus, a flat output of the satellite when we introduce the time-advance operator. More precisely, a flat output of the system over the ring $Alg2[\delta^{-1}]$ is defined by:

```
> evalm([[xi1(t)],[xi2(t)]])=ApplyMatrix(S, [x1(t),x2(t),x3(t),x4(t),u1(t),u2(t)],
> Alg2);
```

$$\begin{bmatrix} \xi 1(t) \\ \xi 2(t) \end{bmatrix} = \begin{bmatrix} -\frac{1}{6}\frac{r\,(\mathrm{D}^{(3)})(x3)(t+1)}{a\,\omega^3\,b} - \frac{2}{3}\frac{r\,\mathrm{D}(x3)(t+1)}{\omega\,a\,b} - \frac{1}{3}\frac{\mathrm{u1}(t)}{\omega^2\,b\,m} + \frac{1}{6}\frac{\mathrm{D}(u2)(t)}{a\,\omega^3\,m} \\ \frac{\mathrm{x3}(t+1)}{b\,a} \end{bmatrix}$$

Using the fact that

$$\begin{aligned} &(x1 : x2 : x3 : x4 : u1 : u2)^T = Ext1[3]\,(\xi 1, \xi 2)^T \\ \text{and}\quad &(\xi 1 : \xi 2)^T = S\,(x1 : x2 : x3 : x4 : u1 : u2)^T, \end{aligned}$$

then we have $(x1 : x2 : x3 : x4 : u1 : u2)^T = Q\,(x1 : x2 : x3 : x4 : u1 : u2)^T$, where Q is the following matrix:

```
> Q := simplify(Mult(Ext1[3], S, Alg2));
```

$$Q := \begin{bmatrix} 0 & 0 & -\frac{r\,Dt\,(Dt^2+4\,\omega^2)}{6\,\omega^3} & 0 & -\frac{a\,\delta}{3\,\omega^2\,m} & \frac{b\,Dt\,\delta}{6\,\omega^3\,m} \\ 0 & 0 & -\frac{Dt^2\,r\,(Dt^2+4\,\omega^2)}{6\,\omega^3} & 0 & -\frac{a\,Dt\,\delta}{3\,\omega^2\,m} & \frac{b\,Dt^2\,\delta}{6\,\omega^3\,m} \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & Dt & 0 & 0 & 0 \\ 0 & 0 & -\frac{m\,r\,Dt^3\,(\omega^2+Dt^2)}{6\,\delta\,a\,\omega^3} & 0 & -\frac{-3\,\omega^2+Dt^2}{3\,\omega^2} & \frac{b\,(-3\,\omega^2+Dt^2)\,Dt}{6\,a\,\omega^3} \\ 0 & 0 & -\frac{Dt^2\,m\,r\,(\omega^2+Dt^2)}{3\,\omega^2\,\delta\,b} & 0 & -\frac{2\,a\,Dt}{3\,\omega\,b} & \frac{Dt^2}{3\,\omega^2} \end{bmatrix}$$

Let us point out that by the form of the matrix S shows that $(\xi 1' : \xi 2')^T = S2\,(x1 : x2 : x3 : x4 : u1 : u2)^T$ is also a flat output of the system, where $S2$ is defined by:

```
> S2 := evalm([[0,0,0,0,-2*omega/b,Dt/a], [0,0,1/(a*b*delta),0,0,0]]);
```

$$S2 := \begin{bmatrix} 0 & 0 & 0 & 0 & -\frac{2\,\omega}{b} & \frac{Dt}{a} \\ 0 & 0 & \frac{1}{\delta\,b\,a} & 0 & 0 & 0 \end{bmatrix}$$

Let us check this by using *OreModules*.

```
> P2 := Factorize(simplify(evalm(delta*Q)),simplify(evalm(delta*S2)), Alg2);
```

$$P2 := \begin{bmatrix} \frac{\delta\, b\, a}{6\,\omega^3\, m} & -\frac{r\, b\, a\, \delta\, Dt^3}{6\,\omega^3} - \frac{2\, r\, b\, a\, \delta\, Dt}{3\,\omega} \\ \frac{b\, a\, \delta\, Dt}{6\,\omega^3\, m} & -\frac{r\, b\, a\, \delta\, Dt^4}{6\,\omega^3} - \frac{2\, r\, b\, a\, \delta\, Dt^2}{3\,\omega} \\ 0 & b\, a\, \delta \\ 0 & b\, a\, \delta\, Dt \\ \frac{b\, Dt^2}{6\,\omega^3} - \frac{b}{2\,\omega} & -\frac{m\, r\, b\, Dt^5}{6\,\omega^3} - \frac{m\, r\, b\, Dt^3}{6\,\omega} \\ \frac{a\, Dt}{3\,\omega^2} & -\frac{m\, r\, a\, Dt^4}{3\,\omega^2} - \frac{a\, Dt^2\, m\, r}{3} \end{bmatrix}$$

Therefore, we have $\delta\, Q = P2\, \delta\, S2$, and thus, $Q = P2\, S2$. Therefore, we obtain

$$\begin{aligned} (x1 : x2 : x3 : x4 : u1 : u2)^T &= Q\,(x1 : x2 : x3 : x4 : u1 : u2)^T \\ &= P2\,(S2\,(x1 : x2 : x3 : x4 : u1 : u2)^T) = P2\,(\xi 1' : \xi 2')^T. \end{aligned}$$

Let us check now that *P2* parametrizes all solutions of the system. If we eliminate the $\xi 1$' and $\xi 2$' from the inhomogeneous system $(x1 : x2 : x3 : x4 : u1 : u2)^T = P2\ (\xi 1' : \xi 2')^T$, then we obtain $R3\,(x1 : x2 : x3 : x4 : u1 : u2)^T = 0$, where the matrix *R3* is defined by:

```
> R3 := SyzygyModule(P2, Alg2);
```

$$R3 := \begin{bmatrix} -3\, m\, \omega^2 & Dt\, m & 0 & -2\, \omega\, r\, m & -a\, \delta & 0 \\ Dt & -1 & 0 & 0 & 0 & 0 \\ 0 & 2\, m\, \omega & 0 & m\, r\, Dt & 0 & -b\, \delta \\ 0 & 0 & Dt & -1 & 0 & 0 \end{bmatrix}$$

```
> Quotient(R2, R3, Alg2);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Therefore, we obtain that the quotient *Alg2*-module (*Alg2*ˆ4 *R2*)/(*Alg2*ˆ4 *R3*) is zero.

```
> Quotient(R3, R2, Alg2);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Moreover, we obtain that the quotient *Alg2*-module (*Alg2*ˆ4 *R3*)/(*Alg2*ˆ4 *R2*) is zero, which proves that the *Alg2*-module associated with *R2* is equal to the *Alg2*-module associated with *R3*. Therefore, $(\xi 1' : \xi 2')^T = S2\,(x1 : x2 : x3 : x4 : u1 : u2)^T$ is also a flat output of the system which satisfies $(x1 : x2 : x3 : x4 : u1 : u2)^T = P2\,(\xi 1' : \xi 2')^T$. Let us notice that the two previous flat outputs only exist for $a\, b \neq 0$.

Let us study the case where $a = 0$ and $b = 1$, i.e., the case where there is only a tangential thrust.

```
> R201 := linalg[submatrix](subs(a=0,b=1,evalm(R2)), 1..4, [1,2,3,4,6]);
```

$$R201 := \begin{bmatrix} Dt & -1 & 0 & 0 & 0 \\ -3\,\omega^2 & Dt & 0 & -2\,\omega\, r & 0 \\ 0 & 0 & Dt & -1 & 0 \\ 0 & \frac{2\,\omega}{r} & 0 & Dt & -\frac{\delta}{m\, r} \end{bmatrix}$$

Let us define a formal adjoint *R201_ adj* of *R201* by using an involution of *Alg2*.

```
> R201_adj := Involution(R201, Alg2):
```

Let us check whether or not the *Alg2*-module associated with *R201* is torsion-free.

```
> Ext101 := Exti(R201_adj, Alg2, 1);
```

$$Ext101 := \left[\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} -3\,\omega^2 & Dt & 0 & -2\,\omega\, r & 0 \\ Dt & -1 & 0 & 0 & 0 \\ 0 & 2\, m\, \omega & 0 & m\, r\, Dt & -\delta \\ 0 & 0 & Dt & -1 & 0 \end{bmatrix}, \begin{bmatrix} 2\,\delta\, Dt\,\omega\, r \\ 2\,\delta\,\omega\, Dt^2\, r \\ -3\,\delta\,\omega^2 + Dt^2\,\delta \\ \delta\, Dt^3 - 3\,\delta\, Dt\,\omega^2 \\ Dt^4\, m\, r + Dt^2\,\omega^2\, m\, r \end{bmatrix} \right]$$

Therefore, we obtain that *Alg2*-module associated with *R201* is torsion-free, and thus, the system associated with R201 is controllable and parametrizable. In particular, a parametrization of the system is given by *Ext101*[3] or, in other words, we have:

```
> Parametrization(R201, Alg2);
```

$$\begin{bmatrix} 2\,\omega\, r\, \mathrm{D}(\xi_1)(t-1) \\ 2\,\omega\, r\, (\mathrm{D}^{(2)})(\xi_1)(t-1) \\ -3\,\omega^2\,\xi_1(t-1) + (\mathrm{D}^{(2)})(\xi_1)(t-1) \\ (\mathrm{D}^{(3)})(\xi_1)(t-1) - 3\,\omega^2\,\mathrm{D}(\xi_1)(t-1) \\ m\, r\, (\mathrm{D}^{(4)})(\xi_1)(t) + m\,\omega^2\, r\, (\mathrm{D}^{(2)})(\xi_1)(t) \end{bmatrix}$$

Let us check whether or not the *Alg2*-module associated with *R201* is free, i.e., whether or not the system associated with *R201* is flat.

```
> Ext201 := Exti(R201_adj, Alg2, 2);
```

$$Ext201 := [\begin{bmatrix} \delta \\ Dt^2\,\omega^2 + Dt^4 \end{bmatrix}, [\ 1\ ], \mathrm{SURJ}(1)]$$

The first matrix is not the identity matrix, and thus, we obtain that the *Alg2*-module associated with *R201* is not a projective, and thus, a free *Alg2*-module by the Quillen-Suslin theorem. The fact that the system is controllable implies that there exists a π-polynomial in δ. Let us compute the minimal one.

```
> PiPolynomial(R201, Alg2, [delta]);
```

$$[\delta]$$

Therefore, if we use the operator δ^{-1}, i.e., an advance operator, then the system becomes flat. Let us compute a flat output of the system which involves the advance operator δ^{-1}. if we use advance operators.

```
> S201 := LocalLeftInverse(Ext101[3], [delta], Alg2);
```

$$S201 := \begin{bmatrix} 0 & \frac{1}{6\,\delta\,\omega^3\, r} & -\frac{1}{3\,\delta\,\omega^2} & 0 & 0 \end{bmatrix}$$

Therefore, $\xi = S201\ (x1 : x2 : x3 : x4 : u2)^T$ is a flat output of the system which satisfies

$$(x1 : x2 : x3 : x4 : u2)^T = Ext101[3]\,\xi.$$

Let us also point out that the flat output of the system is defined by means of an advance operator whereas the parametrization *Ext101*[3] of the system only contains time-delay operators. Let us find a flat output of the system which only use Dt and δ and a parametrization wich depends on Dt, δ and δ^{-1}. In particular, let us prove that $\xi' = T201\ (x1 : x2 : x3 : x4 : u2)^T$, where *T201* is the matrix defined below, is a flat output of the system.

```
> T201 := evalm([[0,1,-2*r*omega,0,0]]);
```

$$T201 := \begin{bmatrix} 0 & 1 & -2\,\omega\,r & 0 & 0 \end{bmatrix}$$

Let us check it. First of all, we have

$$\begin{aligned}(x1 : x2 : x3 : x4 : u2)^T &= Ext101[3]\,\xi = (Ext101[3] \circ S201)\,(x1 : x2 : x3 : x4 : u2)^T \\ &= Q201\,(x1 : x2 : x3 : x4 : u2)^T,\end{aligned}$$

where $Q201$ is defined by:

```
> Q201 := simplify(Mult(Ext101[3], S201, Alg2));
```

$$Q201 := \begin{bmatrix} 0 & \dfrac{Dt}{3\,\omega^2} & -\dfrac{2\,Dt\,r}{3\,\omega} & 0 & 0 \\ 0 & \dfrac{Dt^2}{3\,\omega^2} & -\dfrac{2\,Dt^2\,r}{3\,\omega} & 0 & 0 \\ 0 & -\dfrac{\%1}{6\,\omega^3\,r} & \dfrac{\%1}{3\,\omega^2} & 0 & 0 \\ 0 & -\dfrac{Dt\,\%1}{6\,\omega^3\,r} & \dfrac{Dt\,\%1}{3\,\omega^2} & 0 & 0 \\ 0 & \dfrac{Dt^2\,m\,(Dt^2+\omega^2)}{6\,\omega^3\,\delta} & -\dfrac{Dt^2\,m\,r\,(Dt^2+\omega^2)}{3\,\delta\,\omega^2} & 0 & 0 \end{bmatrix}$$

$$\%1 := 3\,\omega^2 - Dt^2$$

Let us point out that *Q201* is an idempotent of $Alg2^{5\times5}$. This fact can be easily checked:

```
> simplify(evalm(Q201^2 - Q201));
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> P201 := Factorize(evalm(delta*Q201), T201, Alg2);
```

$$P201 := \begin{bmatrix} \dfrac{\delta\,Dt}{3\,\omega^2} \\ \dfrac{\delta\,Dt^2}{3\,\omega^2} \\ -\dfrac{\delta}{2\,\omega\,r} + \dfrac{\delta\,Dt^2}{6\,\omega^3\,r} \\ -\dfrac{\delta\,Dt}{2\,\omega\,r} + \dfrac{\delta\,Dt^3}{6\,\omega^3\,r} \\ \dfrac{Dt^4\,m}{6\,\omega^3} + \dfrac{Dt^2\,m}{6\,\omega} \end{bmatrix}$$

Therefore, we have δ *Q201*= *P201 o T201*, and thus, *Q201* = δ^{-1} *P201 o T201*. Let us denote by *Param* the matrix $\delta^{-1} P201$, namely:

```
> Param := simplify(evalm(delta^(-1)*P201));
```

$$Param := \begin{bmatrix} \frac{Dt}{3\,\omega^2} \\ \frac{Dt^2}{3\,\omega^2} \\ -\frac{3\,\omega^2 - Dt^2}{6\,\omega^3\,r} \\ -\frac{Dt\,(3\,\omega^2 - Dt^2)}{6\,\omega^3\,r} \\ \frac{Dt^2\,m\,(Dt^2 + \omega^2)}{6\,\omega^3\,\delta} \end{bmatrix}$$

Hence, if we define $\xi' = T201\,(x1 : x2 : x3 : x4 : u2)^T$, then for every element $(x1 : x2 : x3 : x4 : u2)^T$ of the system, we have

$$\begin{aligned} (x1 : x2 : x3 : x4 : u2)^T &= Q201\,(x1 : x2 : x3 : x4 : u2)^T = Param\,(T201\,(x1 : x2 : x3 : x4 : u2)^T) \\ &= Param\,\xi'. \end{aligned}$$

Let us study the case where $a = 1$ and $b = 0$, i.e., the case where there is only a radial thrust.

```
> R210 := linalg[submatrix](subs(a=1,b=0,evalm(R2)), 1..4, 1..5);
```

$$R210 := \begin{bmatrix} Dt & -1 & 0 & 0 & 0 \\ -3\,\omega^2 & Dt & 0 & -2\,\omega\,r & -\frac{\delta}{m} \\ 0 & 0 & Dt & -1 & 0 \\ 0 & \frac{2\,\omega}{r} & 0 & Dt & 0 \end{bmatrix}$$

We first define a formal adjoint *R210*_adj of *R201* using an involution of *Alg2*.

```
> R210_adj := Involution(R210, Alg2);
```

$$R210\_adj := \begin{bmatrix} -Dt & -3\,\omega^2 & 0 & 0 \\ -1 & -Dt & 0 & \frac{2\,\omega}{r} \\ 0 & 0 & -Dt & 0 \\ 0 & -2\,\omega\,r & -1 & -Dt \\ 0 & \frac{\delta}{m} & 0 & 0 \end{bmatrix}$$

Let us check whether or not the system defined by the matrix *R210* is controllable and parametrizable.

```
> Ext101 := Exti(R210_adj, Alg2, 1);
```

$$Ext101 := \left[\begin{bmatrix} Dt & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & Dt & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 2\,\omega & 0 & 0 & r & 0 \\ 0 & 2\,\omega & 0 & Dt\,r & 0 \\ 0 & 2\,m\,Dt & 0 & -\omega\,r\,m & -2\,\delta \\ 0 & 0 & Dt & -1 & 0 \end{bmatrix}, \begin{bmatrix} -Dt\,\delta\,r \\ -\delta\,Dt^2\,r \\ 2\,\omega\,\delta \\ 2\,\delta\,Dt\,\omega \\ -Dt\,\omega^2\,r\,m - Dt^3\,r\,m \end{bmatrix} \right]$$

The first matrix of *Ext101* is not the identity matrix, and thus, the *Alg2*-module associated with *R201* is not torsion-free. The torsion elements of this module are defined by:

```
> TorsionElements(R210, [x1(t),x2(t),x3(t),x4(t),u1(t)], Alg2);
```

$$[\begin{bmatrix} \mathrm{D}(\theta_1)(t) = 0 \\ \mathrm{D}(\theta_3)(t) = 0 \end{bmatrix}, \begin{bmatrix} \theta_1(t) = 2\,\omega\,\mathrm{x1}(t) + r\,\mathrm{x4}(t) \\ \theta_3(t) = 2\,m\,\mathrm{D}(x2)(t) - \omega\,r\,m\,\mathrm{x4}(t) - 2\,\mathrm{u1}(t-1) \end{bmatrix}]$$

To finish, the controllable part of the system is defined by *Ext101*[2] and it is parametrized by *Ext101*[3], namely, we have:

```
> ApplyMatrix(Ext101[3], [xi(t)], Alg2);
```

$$\begin{bmatrix} 2\,\omega\,r\,\mathrm{D}(\xi)(t-1) \\ 2\,\omega\,r\,(\mathrm{D}^{(2)})(\xi)(t-1) \\ -3\,\omega^2\,\xi(t-1) + (\mathrm{D}^{(2)})(\xi)(t-1) \\ (\mathrm{D}^{(3)})(\xi)(t-1) - 3\,\omega^2\,\mathrm{D}(\xi)(t-1) \\ r\,m\,(\mathrm{D}^{(4)})(\xi)(t) + \omega^2\,r\,m\,(\mathrm{D}^{(2)})(\xi)(t) \end{bmatrix}$$